Asymptotic energy and wave function of one-electron molecular orbital

Hiroshi Fukuda

School of Administration and Informatics,

University of Shizuoka, Shizuoka-shi, Shizuoka 422, Japan

Abstract

A program for the calculation of one-electron molecular energy and wave function for large inter nuclear distance R is reported.

PROGRAM SUMMARY

Title of program: MOAS

Catalogue number:

Program obtained from: CPC Program Library, Queens University, Belfast, N. Ireland

Computer: FACOM M760/8 and SUN 4/1+; this program is operable on similar computers

Operating system: FACOM OS4/F4 MSP and SunOS 4.1

Programing language used: FORTRAN 77

No. of lines in combined programs and test deck: FORTRAN code file: 513 lines; output file: 121 lines

Keywords: one-electron molecular orbital, asymptotic form

Nature of physical problem

The program calculates asymptotic expansion of the energy and the wave function of one-electron molecular orbital in inverse powers of inter nuclear distance R up to $1/R^N$ term in the energy (up to $1/R^{N-2}$ term in the wave function).

Method of solution

The energy and wave function are obtained exactly by solving the recurrence formula in [1].

Restrictions on the complexity of the problem

The maximum power of $1/R$ in energy, N, is less than or equal to 12, and the principal quantum number of the separated atom, n, is restricted to $n < 55 - 3N$.

Typical running time

The running time of the test run is about 140 sec in Sun 4/1+

References

[1] C. A. Coulson, Proc. Roy. Soc. (Edinburgh) A **61**, 20 (1941); C. A. Coulson and C. M. Gillam, *ibid.* **62**, 360 (1947).

LONG WRITE-UP

1 Method

One-electron diatomic molecular orbital is often used in the field of atomic and molecular physics. Though it is obtained exactly by separation of variables in the confocal elliptic coordinates, accurate numerical calculation for large inter nuclear distance R is difficult in the coordinates. However, Coulson *et al.* [1] derived the formula for its asymptotic expansion of energy and wave function in inverse powers of R. A reported program calculates it, using their formula, up to $1/R^N$ term in the energy (up to $1/R^{N-2}$ term in the wave function) for $N \leq 12$. The energy $E(R)$ and the wave function ψ of the one-electron molecular orbital with nuclear charge Z_A at point A and Z_B at B is defined by the eigenproblem

$$[-\frac{1}{2m}\nabla_\alpha^2 - \frac{Z_A}{r_A} - \frac{Z_B}{r_B} + \frac{Z_A Z_B}{R} - E(R)]\psi = 0. \tag{1}$$

$\mathbf{r}_A$ and $\mathbf{r}_B$ are the position vectors of electron from A and B, respectively, as shown in Fig. 1, ∇_α is the derivative with respect to $\mathbf{r}_\alpha$ ($\alpha = A$ or B), and m is the reduced mass.

For large R, the energy and the wave function in Eq. (1) are expanded in powers of $1/R$

$$E(R) = E_0 + \frac{1}{R}E_1 + \frac{1}{R^2}E_2 + \ldots, \tag{2}$$

$$\psi = \psi_0 + \frac{1}{R}\psi_1 + \frac{1}{R^2}\psi_2 + \ldots\ . \tag{3}$$

E_ρ and ψ_ρ ($\rho = 0, 1, 2, \ldots$) are determined by a set of inhomogeneous perturbation

equations

$$(H_\alpha - E_0)\psi_\rho = -\sum_{l=1}^{\rho}(V_l - E_l)\psi_{\rho-l} \tag{4}$$

with normalization condition

$$\int d\mathbf{r}_\alpha |\psi|^2 = 1 + O(\frac{1}{R^{N-1}}). \tag{5}$$

The unperturbed hamiltonian H_α in Eq. (4), where $\alpha = A$ or B specifies the center to which the wave function concentrates for $R \to \infty$, is

$$H_\alpha = -\frac{1}{2m}\nabla_\alpha^2 - \frac{Z_\alpha}{r_\alpha}, \tag{6}$$

and V_l is the expansion coefficient of the perturbation interaction $V = H - H_\alpha$,

$$V = \frac{1}{R}V_1 + \frac{1}{R^2}V_2 + \dots \ . \tag{7}$$

Eq. (4) is solved exactly in the parabolic coordinates,

$$\begin{aligned} \mu &= r_\alpha(1 + \cos\theta_\alpha) \\ \nu &= r_\alpha(1 - \cos\theta_\alpha), \end{aligned} \tag{8}$$

and ϕ, where θ_α is the angle between $\mathbf{R}$ and $\mathbf{r}_\alpha$, shown in Fig. 1, and ϕ is the azimuthal angle of $\mathbf{r}_\alpha$ around $\mathbf{R}$. The wave function ψ_ρ is obtained in the form [1]

$$\psi_\rho = \frac{1}{Z_\alpha^\rho}\sum_{r=-\rho}^{\rho}\sum_{s=-\rho}^{\rho} c_{r,s}^{(\rho)}\Phi_{r,s}(\frac{mZ_\alpha\mu}{n}, \frac{mZ_\alpha\nu}{n})(\frac{mZ_\alpha}{n}\sqrt{\mu\nu})^{|\lambda|}e^{-\frac{mZ_\alpha(\mu+\nu)}{2n}}\frac{e^{i\lambda\phi}}{\sqrt{2\pi}}. \tag{9}$$

$\Phi_{r,s}(x,y)$ is the product of the confluent hypergeometric function, F, of x and y,

$$\begin{aligned} \Phi_{r,s}(x,y) &= \frac{(k_1 + |\lambda| + r)!(k_2 + |\lambda| + s)!}{(k_1 + r)!(k_2 + s)!(\lambda!)^2} \\ &\quad \times F(-k_1 - r, |\lambda| + 1, x)F(-k_2 - s, |\lambda| + 1, y). \end{aligned} \tag{10}$$

$k_1 = 0, 1, 2, \ldots$ and $k_2 = 0, 1, 2, \ldots$ in Eq. (10) are the parabolic quantum numbers of the unperturbed Hamiltonian H_α, $\lambda = 0, \pm 1, \pm 2, \ldots$ is the azimuthal quantum number, and n is the principal quantum number

$$n = k_1 + k_2 + |\lambda| + 1. \tag{11}$$

The electronic state is specified by a set of quantum numbers $(\alpha, k_1, k_2, \lambda)$ for H_α. Note that, for a symmetric charge system, i.e., $Z_A = Z_B$, the wave function which have correct symmetry is given by $\psi = [\psi(\alpha = A) \pm \psi(\alpha = B)]/\sqrt{2}$ since the electronic energies, $E(R)$, are degenerate with respect to $\alpha = A$ or B in this treatment.

The coefficients $c_{r,s}^{(\rho)}$ $(r, s = -\rho, -\rho+1, \ldots, \rho)$ in Eq. (9) and the energy E_ρ are determined by the following recurrence relations.

$$\begin{aligned}
E_\rho &= -\frac{Z_\beta}{2nZ_\alpha^{\rho-1}}[T_{0,0}^{(\rho)}(0,0) \\
&\quad + \frac{1}{c_{0,0}^{(0)}} \sum_{p+q\neq 0} c_{p,q}^{(\rho-2)} K_{p,q}^{(2)}(0,0) + \frac{1}{c_{0,0}^{(0)}} \sum_{l=3}^{\rho-1} \sum_{p,q} c_{p,q}^{(\rho-l)} K_{p,q}^{(l)}(0,0)], \qquad (12) \\
c_{r,-r}^{(\rho-2)} &= -\frac{1}{A_{r,-r} + B_{r,-r}}[\sum_{p+q\neq 0} c_{p,q}^{(\rho-2)} K_{p,q}^{(2)}(r,-r) \\
&\quad + \sum_{l=3}^{\rho} \sum_{p,q} c_{p,q}^{(\rho-l)} K_{p,q}^{(l)}(r,-r)] \ \text{ for } r \neq 0, \qquad (13) \\
c_{0,0}^{(\rho-2)} &= -[c_{0,0}^{(0)}]^2 \sum_{(p,q)\neq(0,0)} c_{p,q}^{(\rho-2)} L(0,0;p,q) \\
&\quad - \frac{c_{0,0}^{(0)}}{2} \sum_{k=1}^{\rho-3} \sum_{p,q} \sum_{p',q'} c_{p,q}^{(k)} c_{p',q'}^{(\rho-2-k)} L(p,q;p',q'), \qquad (14) \\
c_{r,s}^{(\rho)} &= \frac{Z_\beta n^2}{2mZ_\alpha(r+s)}[c_{-s,s}^{(\rho-2)} A_{r,s} + c_{r,-r}^{(\rho-2)} B_{r,s} + \sum_{p+q\neq 0} c_{p,q}^{(\rho-2)} K_{p,q}^{(2)}(r,s) \\
&\quad + \sum_{l=3}^{\rho} \sum_{p,q} c_{p,q}^{(\rho-l)} K_{p,q}^{(l)}(r,s)] \ \text{ for } r+s \neq 0, \qquad (15)
\end{aligned}$$

where the summations over p, q and p', q' are restricted by the condition,

$$c_{p,q}^{(\rho)} = 0 \text{ for } |p| > \rho \text{ or } |q| > \rho, \tag{16}$$

and

$$\begin{aligned}
K_{p,q}^{(l)}(r,s) &= \frac{Z_\alpha^{l-1} E_l}{Z_\beta}(a_{p,r-p}^{(1)}(k_1)a_{q,s-q}^{(0)}(k_2) + a_{p,r-p}^{(0)}(k_1)a_{q,s-q}^{(1)}(k_2)) \\
&\quad + T_{r-p,s-q}^{(l)}(r,s), \qquad (17)\\
T_{p,q}^{(l)}(r,s) &= \sum_{k=0}^{l} v_k^{(l)} a_{r-p,p}^{(l-k)}(k_1) a_{s-q,q}^{(k)}(k_2), \qquad (18)\\
L(p,q;p',q') &= \frac{1}{4}(\frac{n}{mZ_\alpha})^3 \frac{(k_1+|\lambda|+p)!(k_2+|\lambda|+q)! a_{p,0}^{(0)}(k_1) a_{q,0}^{(0)}(k_2)}{(k_1+p)!(k_2+q)!} \\
&\quad \times \sum_{i=-1}^{1} [a_{p',i}^{(1)}(k_1)\delta_{p,p'+i}\delta_{q,q'} + a_{q',i}^{(1)}(k_2)\delta_{p,p'}\delta_{q,q'+i}], \qquad (19)\\
A_{r,s} &= \frac{Z_\alpha E_2}{Z_\beta} a_{-s,r+s}^{(1)}(k_1) + v_0^{(2)} a_{-s,r+s}^{(2)}(k_1), \qquad (20)\\
B_{r,s} &= \frac{Z_\alpha E_2}{Z_\beta} a_{-r,r+s}^{(1)}(k_2) - v_0^{(2)} a_{-r,r+s}^{(2)}(k_2), \qquad (21)\\
a_{r,s}^{(p)}(k) &= \begin{cases}
(-)^s \dfrac{(p!)^2 (k+r+s)!}{(k+|\lambda|+r+s)!} \\
\quad \times \displaystyle\sum_j \frac{(p+|\lambda|+k+r-j)!}{(j+s)!j!(p-s-j)!(p-j)!(k+r-j)!} \\
\quad \text{for } \max(-k-r,-p) \le s \le p \text{ and } r \ge -k \\
0 \quad \text{otherwise.}
\end{cases} \qquad (22)
\end{aligned}$$

$v_k^{(\rho)}$ is defined by the relation

$$\sum_{k=0}^{\rho} v_k^{(\rho)} \mu^{\rho-k} \nu^k = -\frac{\mu+\nu}{Z_\beta}(\frac{n}{m})^{\rho-1} V_\rho. \tag{23}$$

The summation in Eq. (22) is over all integral values of j which make every one of the factorials non-negative. The values of $v_k^{(\rho)}$ for $\rho \le 12$ are tabulated in Table 1.

Suppose $\{E_0, E_1, \ldots, E_{\rho-1}\}$, $\{c_{r,s}^{(0)}, \ldots, c_{r,s}^{(\rho-3)}$ for all r and $s\}$, and $\{c_{r,s}^{(\rho-2)}$ for $r+s \neq 0\}$ are given, then Eqs. (12)–(15) determine E_ρ, $c_{r,-r}^{(\rho-2)}$ for all r, and $c_{r,s}^{(\rho)}$ for $r+s \neq$

0. Thus, for a given N, $\{E_0, E_1, \ldots, E_N\}$ and $\{\psi_0, \psi_1, \ldots, \psi_{N-2}\}$ are calculated successively by Eqs. (12)–(15) with initial conditions

$$E_0 = -\frac{mZ_\alpha}{2n^2}, \quad E_1 = (Z_\alpha - 1)Z_\beta, \quad E_2 = -\frac{Z_\beta}{2Z_\alpha n} T^{(2)}_{0,0}(0,0), \tag{24}$$

and

$$\begin{aligned} c^{(0)}_{0,0} &= \sqrt{\frac{2(mZ_\alpha)^3 k_1! k_2!}{n^2 (k_1 + |\lambda|)! (k_2 + |\lambda|)!}} \\ c^{(1)}_{r,s} &= 0 \ \text{ for } r + s \neq 0, \\ c^{(2)}_{r,s} &= \frac{1}{r+s} \frac{Z_\beta n^2 c^{(0)}_{0,0}}{2mZ_\alpha} [\frac{Z_\beta E_2}{Z_\alpha} (a^{(1)}_{0,r} b^{(0)}_{0,s} + a^{(0)}_{0,r} b^{(1)}_{0,s}) + T^{(2)}_{r,s}(r,s)] \\ &\quad \text{for } r + s \neq 0. \end{aligned} \tag{25}$$

In the above equations, $Z_\beta = Z_B$ for $\alpha = A$ and $Z_\beta = Z_A$ for $\alpha = B$.

It may be convenient to rewrite Eq. (9) in the form

$$\psi_\rho = \sum_{r=0}^{\rho+k_1} \sum_{s=0}^{\rho+k_2} d^{(\rho)}_{r,s} \mu^r \nu^s (\mu\nu)^{|\lambda|/2} e^{-\frac{mZ_\alpha(\mu+\nu)}{2n}} \frac{e^{i\lambda\phi}}{\sqrt{2\pi}}. \tag{26}$$

for the purpose of numerical use of ψ.

2 Description of the program

The subroutine MOAS calculates the energy E_ρ ($\rho = 0, 1, \ldots, N$), the coefficients $c^{(\rho)}_{r,s}$ ($\rho = 0, 1, \ldots, N-2$) in Eq. (9), and $d^{(\rho)}_{r,s}$ ($\rho = 0, 1, \ldots, N-2$) in Eq. (26) for a given set of quantum numbers $(\alpha, k_1, k_2, \lambda)$. The restrictions of the program are

$$N \leq 12 \tag{27}$$

and

$$n \leq M - 3N \tag{28}$$

where n is the principal quantum number given by Eq. (11) and M is the maximum value of which factorial can be treated by a computer. For FACOM M series, $M = 55$.

The user can check that the output is correct by comparing the energy to the exact expressions (24) for E_0 and E_1, and

$$\begin{aligned} E_2 &= -\frac{3Z_\beta n\Delta}{2mZ_\alpha}, \\ E_3 &= \frac{Z_\beta n^2}{2m^2 Z_\alpha^2}(n^2 - 6\Delta^2 - 1), \\ E_4 &= \frac{Z_\beta^2 n^4}{16m^3 Z_\alpha^4}(-17n^2 + 9\lambda^2 + 3\Delta^2 - 19) \\ &\quad + \frac{Z_\beta n^3 \Delta}{16m^3 Z_\alpha^3}(39n^2 + 9\lambda^2 - 109\Delta^2 - 59), \end{aligned} \tag{29}$$

where $\Delta = k_1 - k_2$ for $\alpha = A$ and $\Delta = k_2 - k_1$ for $\alpha = B$. Also the orthogonality relation of the wave functions following Eq. (5),

$$\sum_{l=0}^{\rho} \langle \psi_l | \psi_{\rho-l} \rangle = \delta_{0,\rho} \ \text{ for } \ 0 \le \rho \le N-2 \tag{30}$$

can be used to see the wave functions are calculated correctly. The left hand side of Eq. (30) is given by the array OL(ρ) (see the description of output below).

Input

data:	ZA, ZB, AM0, NA, KK1, KK2, MM, NR, NC, ND
description:	Nuclear charge Z_A and Z_B, mass parameter m, quantum numbers $\alpha, k_1, k_2, \lambda$, and N. NC and ND are the size of arrays for EV, CC and DD (see below).

Output

data: EV, CC, DD, OL, OX

description: EV(I)= E_i, CC(L, I, J)= $c_{i,j}^{(l)}$ and DD(L, I, J)= $d_{i,j}^{(l)}$. OL's give the norm of ψ as

$$\langle\psi|\psi\rangle = \mathrm{OL}(0) + \mathrm{OL}(1)/R + \cdots + \mathrm{OL}(2N-4)/R^{2N-4}$$

and OX(L)= $\langle\psi_l|\psi_l\rangle$.

Subroutines and functions

SETVV: calculation of $v_k^{(\rho)}$ in Eq. (23)

COEFO: calculation of $L(r,s;r',s')$ in Eq. (19)

COEFT: calculation of $T_{p,q}^{(l)}(r,s)$ in Eq. (18)

COEFA: calculation of $a_{r,s}^{(p)}$ in Eq. (22)

COEFH: calculation of coefficients of z^i term of hyper geometric function $F(a,b,z)$

SETFAC: calculation of factorial

Common blocks

COMPOT: contains the array VV(I, J)=$v_j^{(i)}$

FACTRL: contains the array FAC(N)=$n!$

3 Test run

The test run presents the $\{E_0, E_1, \ldots, E_{10}\}$ and $\{d_{r,s}^{(0)}, d_{r,s}^{(1)}, \ldots, d_{r,s}^{(4)}\}$ for three electronic states, $(\alpha, k_1, k_2, \lambda) = (A,0,0,0)$, $(A,1,0,0)$, $(A,0,0,1)$, of antiprotonic Helium atom, $\bar{p}\mathrm{He}^+$ ($Z_A = 2$, $Z_B = -1$).

References

[1] C. A. Coulson, Proc. Roy. Soc. (Edinburgh) A **61**, 20 (1941); C. A. Coulson and C. M. Gillam, *ibid.* **62**, 360 (1947).

TABLES

Table 1. $v_k^{(\rho)}$ in Eq. (23) for $\rho \leq 12$. $v_k^{(\rho)}$ for $k \leq \rho/2$ are obtained by multiplying $(-2m\kappa_\alpha/n)^{1-\rho}$ to tabulated value and $v_k^{(\rho)}$ for $k > \rho/2$ by the relation $v_{\rho-k}^{(\rho)} = \kappa_\alpha^{\rho-1} v_k^{(\rho)}$, where $\kappa_A = -1$ and $\kappa_B = +1$.

	k						
ρ	0	1	2	3	4	5	6
1	1	1					
2	1	0					
3	1	-3					
4	1	-8	0				
5	1	-15	20				
6	1	-24	75	0			
7	1	-35	189	-175			
8	1	-48	392	-784	0		
9	1	-63	720	-2352	1564		
10	1	-80	1215	-5760	8820	0	
11	1	-99	1925	-12375	29700	-19404	
12	1	-120	2904	-24200	81675	-104544	0

FIGURE CAPTIONS

Fig. 1 Coordinates.


e
$\mathbf{r}_A$
$\mathbf{r}_B$
θ_A
θ_B
A
R
B

TEST RUN OUTPUT

```
SYSTEM: ZA= 2.0, ZB=-1.0, MASS=1.0
STATE:  (ALPHA,K1,K2,LAMBDA)=(  1  0  0  0)

E( 0)= -2.00000000000D+00    E( 1)= -1.00000000000D+00
E( 2)=  0.00000000000D+00    E( 3)=  0.00000000000D+00
E( 4)= -1.40625000000D-01    E( 5)=  0.00000000000D+00
E( 6)= -1.17187500000D-01    E( 7)= -2.19404296875D+01
E( 8)= -3.10592651367D-01    E( 9)=  8.65722656250D-01
E(10)= -3.11417999268D+01

D(0)
 R          S= 0
 0   4.00000D+00

D(1)
 R          S= 0             1
 0   0.00000D+00   0.00000D+00
 1   0.00000D+00   0.00000D+00

D(2)
 R          S= 0             1             2
 0   0.00000D+00   5.00000D-01   2.50000D-01
 1  -5.00000D-01   0.00000D+00   0.00000D+00
 2  -2.50000D-01   0.00000D+00   0.00000D+00

D(3)
 R          S= 0             1             2             3
 0   0.00000D+00   0.00000D+00  -1.25000D-01  -8.33333D-02
 1   0.00000D+00   5.00000D-01   2.50000D-01   0.00000D+00
 2  -1.25000D-01   2.50000D-01   0.00000D+00   0.00000D+00
 3  -8.33333D-02   0.00000D+00   0.00000D+00   0.00000D+00

D(4)
 R          S= 0             1             2             3             4
 0  -4.84375D-01   0.00000D+00   8.59375D-02   8.33333D-02   3.90625D-02
 1   0.00000D+00  -6.25000D-02  -4.06250D-01  -2.50000D-01   0.00000D+00
 2   8.59375D-02   3.43750D-01  -1.56250D-02   0.00000D+00   0.00000D+00
 3   0.00000D+00   2.50000D-01   0.00000D+00   0.00000D+00   0.00000D+00
 4  -2.34375D-02   0.00000D+00   0.00000D+00   0.00000D+00   0.00000D+00


SYSTEM: ZA= 2.0, ZB=-1.0, MASS=1.0
```

```
STATE:  (ALPHA,K1,K2,LAMBDA)=(  1  1  0  0)

E( 0)= -5.00000000000D-01    E( 1)= -1.00000000000D+00
E( 2)=  1.50000000000D+00    E( 3)=  1.50000000000D+00
E( 4)= -4.50000000000D+00    E( 5)= -3.82500000000D+01
E( 6)= -1.50000000000D+02    E( 7)= -1.37531250000D+04
E( 8)=  7.87031250000D+03    E( 9)=  6.62630625000D+04
E(10)= -4.43932781250D+05

D(0)
 R          S= 0
 0   1.00000D+00
 1  -1.00000D+00

D(1)
 R          S= 0             1
 0  -5.00000D-01   5.00000D-01
 1   0.00000D+00   0.00000D+00
 2   0.00000D+00   0.00000D+00

D(2)
 R          S= 0             1             2
 0   6.25000D-01   1.25000D+00   1.25000D-01
 1  -1.87500D+00  -1.25000D+00  -1.25000D-01
 2   1.25000D-01   0.00000D+00   0.00000D+00
 3   1.25000D-01   0.00000D+00   0.00000D+00

D(3)
 R          S= 0             1             2             3
 0   5.62500D-01   2.81250D+00   5.62500D-01   2.08333D-02
 1  -3.37500D+00  -2.62500D+00   1.25000D-01   4.16667D-02
 2   5.62500D-01  -6.87500D-01  -1.25000D-01   0.00000D+00
 3   2.08333D-01  -1.25000D-01   0.00000D+00   0.00000D+00
 4   4.16667D-02   0.00000D+00   0.00000D+00   0.00000D+00

D(4)
 R          S= 0             1             2             3             4
 0  -2.50391D+01   4.59375D+00   2.14062D+00   1.56250D-01   2.60417D-03
 1   2.04453D+01  -6.40625D+00  -7.34375D-01  -1.97917D-01  -2.34375D-02
 2   2.14062D+00  -1.40625D+00   4.53125D-01   1.25000D-01   0.00000D+00
 3   3.17708D-01  -3.64583D-01   1.56250D-02   0.00000D+00   0.00000D+00
 4   2.34375D-02  -1.25000D-01   0.00000D+00   0.00000D+00   0.00000D+00
 5   7.81250D-03   0.00000D+00   0.00000D+00   0.00000D+00   0.00000D+00
```

```
SYSTEM: ZA= 2.0, ZB=-1.0, MASS=1.0
STATE:  (ALPHA,K1,K2,LAMBDA)=(  1  0  0  1)

E( 0)= -5.00000000000D-01    E( 1)= -1.00000000000D+00
E( 2)=  0.00000000000D+00    E( 3)= -1.50000000000D+00
E( 4)= -4.87500000000D+00    E( 5)=  0.00000000000D+00
E( 6)=  3.75000000000D+01    E( 7)= -1.26251250000D+04
E( 8)= -1.37062500000D+02    E( 9)= -1.79287500000D+03
E(10)= -6.95002500000D+05

D(0)
 R          S= 0
 0   1.00000D+00

D(1)
 R          S= 0             1
 0   0.00000D+00   0.00000D+00
 1   0.00000D+00   0.00000D+00

D(2)
 R          S= 0             1             2
 0   0.00000D+00   7.50000D-01   1.25000D-01
 1  -7.50000D-01   0.00000D+00   0.00000D+00
 2  -1.25000D-01   0.00000D+00   0.00000D+00

D(3)
 R          S= 0             1             2             3
 0  -4.00000D+00   0.00000D+00  -1.25000D-01  -4.16667D-02
 1   0.00000D+00   7.50000D-01   1.25000D-01   0.00000D+00
 2  -1.25000D-01   1.25000D-01   0.00000D+00   0.00000D+00
 3  -4.16667D-02   0.00000D+00   0.00000D+00   0.00000D+00

D(4)
 R          S= 0             1             2             3             4
 0  -2.38750D+01  -2.25000D+00   2.18750D-01   1.35417D-01   2.34375D-02
 1   2.25000D+00  -5.62500D-01  -5.93750D-01  -1.25000D-01   0.00000D+00
 2   9.68750D-01   4.06250D-01  -1.56250D-02   0.00000D+00   0.00000D+00
 3   9.37500D-02   1.25000D-01   0.00000D+00   0.00000D+00   0.00000D+00
 4  -7.81250D-03   0.00000D+00   0.00000D+00   0.00000D+00   0.00000D+00
```

LISTING OF PROGRAM SOURCE

```
C
C     MAIN PROGRAM FOR TEST RUN
C
C     H. FUKUDA  1993/4/5
C
      IMPLICIT REAL*8(A-H,O-Z)
C
C     ARRAY SIZE
      PARAMETER(NC=12,ND=12)
C
      DIMENSION
C     FOR STATES
     &   NA(10),K1(10),K2(10),M(10)
C     FOR SUBROUTINE MOAS
     &  ,EV(0:NC),CC(0:NC,-NC:NC,-NC:NC),DD(0:NC,0:ND,0:ND)
     &  ,OL(0:2*NC),OX(0:NC)
C
      DATA
C     PARABOLIC QUANTUM NUMBERS
     &   (NA(I),K1(I),K2(I),M(I),I=1,3)
     &   /1,0,0,0,  1,1,0,0,  1,0,0,1/
C     NUCLEAR CHARGE, REDUCED MASS
C      AND THE HIGHEST POWER OF 1/R FOR ENERGY
     &   ZA,ZB,AM0,NR /2.D0, -1.D0, 1.D0, 10/
C
      DO 20 K=1,3
C     LOOP FOR STATES
         NNA=NA(K)
         KK1=K1(K)
         KK2=K2(K)
         MM=M(K)
         CALL MOAS(ZA,ZB,AM0,NNA,KK1,KK2,MM,
     &             EV,CC,DD,OL,OX,NR,NC,ND)
C
         WRITE(6,100) ZA,ZB,AM0,NNA,KK1,KK2,MM,(I,EV(I),I=0,NR)
         DO 10 L=0,4
            WRITE(6,101) 'D',L,(J,J=0,KK2+L)
         DO 14 I=0,KK1+L
            WRITE(6,102) I,(DD(L,I,J),J=0,KK2+L)
   14    CONTINUE
   10    CONTINUE
   20 CONTINUE
```

```
C
  100 FORMAT(//1X,'SYSTEM: ZA=',F4.1,', ZB=',F4.1,', MASS=',F3.1
     &        /1X,'STATE:  (ALPHA,K1,K2,LAMBDA)=(',4I3,')'
     &      //(1X,2('E(',I2,')=',1PD19.11,4X:)))
  101 FORMAT(/1X,A,'(',I1,')'/1X,' R',10X,'S=',I2,7I14)
  102 FORMAT(1X,I2,1P8D14.5)
      STOP
      END
C=============================================
      SUBROUTINE MOAS(ZA,ZB,AM0,NA,KK1,KK2,MM,
     &                EV,CC,DD,OL,OX,NR,NC,ND)
C=============================================
C     ASYMPTOTIC FORM OF ENERGY AND WAVE FUNCTION FOR
C     ONE-ELECTRON MOLECULAR ORBITAL
C
C     BY H. FUKUDA  1993/4/5
C
C    -INPUT-
C     AM0,ZA,ZB: REDUCED MASS, CHARGE OF A AND B.
C     NA:    =1 (ZA,E) + ZB,  =2 (ZB,E) + ZA
C     KK1,KK2,MM: PARABOLIC QUANTUM NUMBERS
C     NR:    ORDER YOU ARE NECCESARY
C     NC,ND: SIZE OF ARRAY.
C
C    -OUTPUT-
C     EV(L): ENERGY E, L=0,...,NR, WHICHI IS GIVEN BY
C            E=EV(0) + EV(1)/R + ...
C            WHERE R IS INTER NUCLEAR DISTANCE.
C     CC(L,I,J): WAVE FUNCTION PSI IS GIVEN BY
C                PSI=PSI(0) + PSI(1)/R + ... + PSI(NR-2)/R**(NR-2)
C                PSI(L)=SUM_(I=-L,L,J=-L,L)
C                           CC(L,I,J)*((I,J))/ZC**L
C                          *EXP(I*MM*PHI)/SQRT(2*PI)
C            WHERE
C              PP=AM0*ZC/N, N=K1+K2+ABS(M)+1
C              ((I,J))= W(KK1+I,PP*X)*W(KK2+J,PP*Y)
C              W(K,Z)=(K+MM)!/(K!MM!)*F(-K,MM+1,Z)*Z**(MM/2)*EXP(-Z/2)
C            F IS THE CONFLUENT HYPER GEOMETRIC FUNCTION,
C            X,Y THE PARABOLIC COORDINATES DEFINED BY
C               X=S+Z, Y=S-Z
C               (S THE DISTANCE FROM NUCLEUS TO ELECTRON AND
C                Z THE PROJECTION OF S ONTO R)
C            AND ZC=ZA OR ZB ACCORDING TO NA.
C     DD(L,I,J): WAVE FUNCTION PSI IS GIVEN BY
```

```
C               PSI(L)=SUM_(I=0,KK1+L,J=0,KK2+L)
C                         DD(L,I,J)*X**I*Y**J*(X*Y)**MM
C                         *EXP(I*MM*PHI)/SQRT(2*PI)
C     OL(L): NORM OF THE WAVE FUNCTION. L=0,...,2*NR-4
C             <PSI PSI>= OL(0)+OL(1)/R+...+OL(2*NR-4)/R**(2*NR-4)
C     OX(L): NORM OF THE WAVE FUNCTION <PSI(L) PSI(L)>
C
      IMPLICIT REAL*8(A-H,O-Z)
C     ARRAY SIZE FOR INTERNAL COMMON BLOCKS
C     COMMON BLOCK FOR VV
      PARAMETER( NV=12 )
      COMMON /COMPOT/VV(0:NV,0:NV)
C     COMMON BLOCK FOR FACTORIAL
      PARAMETER( NFAC=55 )
      COMMON /FACTRL/FAC(0:NFAC)
C
      DIMENSION
     & EV(0:NC),CC(0:NC,-NC:NC,-NC:NC),DD(0:NC,0:ND,0:ND)
     &,OL(0:2*NC),OX(0:NC)
C
C     FUNCTIONS
      A2(I,J)=    EV(2)*COEFA(1,-J,I+J,KK1,MM)
     &         +VV(2,0)*COEFA(2,-J,I+J,KK1,MM)
      B2(I,J)=    EV(2)*COEFA(1,-I,I+J,KK2,MM)
     &         -VV(2,0)*COEFA(2,-I,I+J,KK2,MM)
      SS(I,J,IR,IS)=COEFA(1,IR-I,I,KK1,MM)*COEFA(0,IS-J,J,KK2,MM)
     &             +COEFA(0,IR-I,I,KK1,MM)*COEFA(1,IS-J,J,KK2,MM)
      ST(L,IP,IQ,IR,IS)=COEFT(L,IP,IQ,IR,IS,KK1,KK2,MM)
      SK(L,IP,IQ,IR,IS)=EV(L)*SS(IR-IP,IS-IQ,IR,IS)
     &                 +COEFT(L,IR-IP,IS-IQ,IR,IS,KK1,KK2,MM)
C
C     CHECK INPUT DATA
      MA=ABS(MM)
      NN=KK1+KK2+MA+1
      IF(NR.LT.2)        STOP '*** NR<2, IN MOAS'
      IF(NR.GT.NC)       STOP '*** NR>NC, IN MOAS'
      IF(KK1+NR-2.GT.ND) STOP '*** K1+NR-2>ND, IN MOAS'
      IF(KK2+NR-2.GT.ND) STOP '*** K2+NR-2>ND, IN MOAS'
      IF(NR.GT.NV)       STOP '*** NR>NV, IN MOAS'
C
      IF(KK1.LT.0)               STOP '*** K1<0, IN MOAS'
      IF(KK2.LT.0)               STOP '*** K2<0, IN MOAS'
      IF(NA.LT.1.OR.NA.GT.2)     STOP '*** NA<1 OR NA>2, IN MOAS'
      IF(ZA.LT.0.D0.AND.NA.EQ.1) STOP '*** ZA<0 & NA=1, IN MOAS'
```

```
      IF(ZB.LT.0.D0.AND.NA.EQ.2) STOP '*** ZB<0 & NA=1, IN MOAS'
      IF(3*NR+NN.GT.NFAC)        STOP '*** 3*NR+N>NFAC, IN MOAS'
C
C     PARAMETERS
      IF(NA.EQ.1) THEN
         ZZA=ZA
         ZZB=ZB
      ELSE
         ZZA=ZB
         ZZB=ZA
      END IF
C
      OMG=ZZB/ZZA/AM0
      PP=AM0*ZZA/NN
C
C     FACTORIAL, VV
      IF(FAC(0).NE.1.D0) CALL SETFAC
      CALL SETVV(ZZA,AM0,NN,NA)
C
C     CLEAR CC, DD
      DO 12 L=0,NR
      DO 13 I=-NR,NR
      DO 13 J=-NR,NR
         CC(L,I,J)=0
   13 CONTINUE
      DO 14 I=0,ND
      DO 14 J=0,ND
         DD(L,I,J)=0
   14 CONTINUE
   12 CONTINUE
C
C     0TH ORDER EQUATION.
      EV(0)=-0.5D0*AM0*ZZA**2/NN**2/(ZZA*ZZB)
      CC(0,0,0)=(ZZA*AM0)**1.5D0*SQRT(2.D0)/NN**2
     &          *SQRT( FAC(KK1)/FAC(KK1+MA)*FAC(KK2)/FAC(KK2+MA) )
C    &          *(-1)**(NA*MA)
C
C     1ST ORDER EQUATION.
      EV(1)=-VV(1,0)
C
C     2ND ORDER EQUATION.
      EV(2)=-0.5D0/NN*ST(2,0,0,0,0)
      DO 20 I=-2,2
      DO 20 J=-2,2
```

```
         IF(I+J.EQ.0) GOTO 20
         IF(KK1+I.LT.0 .OR. KK2+J.LT.0) GOTO 20
         CC(2,I,J)=0.5D0/(I+J)*OMG*NN**2*CC(0,0,0)*SK(2,0,0,I,J)
   20 CONTINUE
C
C     LOOP FOR N'TH ORDER EQUATION. N=3,4,5,...,NR
C
      DO 40 N=3,NR
         L=N-2
         SUM=0
         DO 42 I=-L,L
         DO 42 J=-L,L
            IF(I+J.EQ.0) GOTO 42
            IF(KK1+I.LT.0 .OR. KK2+J.LT.0) GOTO 42
            SUM=SUM+CC(L,I,J)*SK(2,I,J,0,0)
   42    CONTINUE
         DO 44 L=3,N-1
         DO 44 I=-N+L,N-L
         DO 44 J=-N+L,N-L
            IF(KK1+I.LT.0 .OR. KK2+J.LT.0) GOTO 44
            SUM=SUM+CC(N-L,I,J)*SK(L,I,J,0,0)
   44    CONTINUE
         SUM=SUM/CC(0,0,0)
         EV(N)=-0.5D0/NN*(ST(N,0,0,0,0)+SUM)
C
         DO 50 I=-N+2,N-2
            IF(I.EQ.0) GOTO 50
            IF(KK1+I.LT.0 .OR. KK2-I.LT.0) GOTO 50
            SUM=0
            DO 52 II=-N+2,N-2
            DO 52 JJ=-N+2,N-2
               IF(II+JJ.EQ.0) GOTO 52
               IF(KK1+II.LT.0 .OR. KK2+JJ.LT.0) GOTO 52
               SUM=SUM+CC(N-2,II,JJ)*SK(2,II,JJ,I,-I)
   52       CONTINUE
            DO 54 L=3,N
            DO 54 II=-N+L,N-L
            DO 54 JJ=-N+L,N-L
               IF(KK1+II.LT.0 .OR. KK2+JJ.LT.0) GOTO 54
               SUM=SUM+CC(N-L,II,JJ)*SK(L,II,JJ,I,-I)
   54       CONTINUE
            CC(N-2,I,-I)=-SUM/(A2(I,-I)+B2(I,-I))
   50    CONTINUE
C
```

```
      L=N-2
      SUM1=0
      DO 46 IR=-L,L
      DO 46 IS=-L,L
         IF(IR.EQ.0.AND.IS.EQ.0) GOTO 46
         IF(KK1+IR.LT.0  .OR. KK2+IS.LT.0) GOTO 46
         SUM1=SUM1+CC(L,IR,IS)*COEFL(0,0,IR,IS,PP,KK1,KK2,MA)
46    CONTINUE
      SUM2=0
      DO 48 K=1,L-1
      DO 48 IR=-K,K
      DO 48 IS=-K,K
      DO 48 IRP=-L+K,L-K
      DO 48 ISP=-L+K,L-K
         IF(KK1+IR.LT.0  .OR. KK2+IS.LT.0 .OR.
     &       KK1+IRP.LT.0 .OR. KK2+ISP.LT.0) GOTO 48
         SUM2=SUM2+CC(K,IR,IS)*CC(L-K,IRP,ISP)
     &              *COEFL(IR,IS,IRP,ISP,PP,KK1,KK2,MA)
48    CONTINUE
      CC(N-2,0,0)=-CC(0,0,0)**2*SUM1-0.5D0*CC(0,0,0)*SUM2
C
      DO 60 I=-N,N
      DO 60 J=-N,N
         IF(I+J.EQ.0) GOTO 60
         IF(KK1+I.LT.0 .OR. KK2+J.LT.0) GOTO 60
         SUM=0.D0
         DO 62 II=-N+2,N-2
         DO 62 JJ=-N+2,N-2
            IF(II+JJ.EQ.0) GOTO 62
            IF(KK1+II.LT.0 .OR. KK2+JJ.LT.0) GOTO 62
            SUM=SUM+CC(N-2,II,JJ)*SK(2,II,JJ,I,J)
62       CONTINUE
         DO 64 L=3,N
         DO 64 II=-N+L,N-L
         DO 64 JJ=-N+L,N-L
            IF(KK1+II.LT.0 .OR. KK2+JJ.LT.0) GOTO 64
            SUM=SUM+CC(N-L,II,JJ)*SK(L,II,JJ,I,J)
64       CONTINUE
         CC(N,I,J)=0.5D0/DBLE(I+J)*OMG*NN**2
     &              *( CC(N-2,-J,J)*A2(I,J)+CC(N-2,I,-I)*B2(I,J)
     &                +SUM )
60    CONTINUE
40 CONTINUE
C
```

```
C     END OF LOOP
C
      DO 90 L=0,NR
         EV(L)=EV(L)*ZZB/ZZA**(L-1)
   90 CONTINUE
C
C     CLEAR CC UNUSED
      DO 94 L=NR-1,NR
      DO 94 I=-NR,NR
      DO 94 J=-NR,NR
         CC(L,I,J)=0
   94 CONTINUE
C
C     COEFFICIENT DD(L,I,J) OF MU**I*NU**J
      DO 70 L=0,NR-2
      DO 70 I=-KK1,L
      DO 70 J=-KK2,L
      DO 70 II=0,KK1+I
      DO 70 JJ=0,KK2+J
         CALL COEFH(-KK1-I,1+MA,F1,II)
         CALL COEFH(-KK2-J,1+MA,F2,JJ)
         DD(L,II,JJ)=DD(L,II,JJ)
     &          +FAC(KK1+MA+I)/FAC(KK1+I)
     &          *FAC(KK2+MA+J)/FAC(KK2+J)/FAC(MA)**2
     &          *CC(L,I,J)/ZZA**L*F1*F2*PP**(II+JJ+MA)
   70 CONTINUE
C
C     NORM OF THE L'TH ORDER OF WAVE FUNCTION
      DO 97 L=0,NR-2
         OX(L)=0
      DO 97 I=-L,L
      DO 97 J=-L,L
      DO 97 IP=-L,L
      DO 97 JP=-L,L
         OX(L)=OX(L)+CC(L,I,J)*CC(L,IP,JP)*1/ZZA**(2*L)
     &           *COEFL(I,J,IP,JP,PP,KK1,KK2,MA)
   97 CONTINUE
C
C     NORM OF THE WAVE FUNCTION
      DO 99 L=0,2*(NR-2)
         OL(L)=0
      DO 99 IL=0,L
      DO 99 I=-IL,IL
      DO 99 J=-IL,IL
```

```
      DO 99 IP=-L+IL,L-IL
      DO 98 JP=-L+IL,L-IL
         IF(IL.GT.NR-2.OR.L-IL.GT.NR-2) GOTO 98
         OL(L)=OL(L)+CC(IL,I,J)*CC(L-IL,IP,JP)*1/ZZA**L
     &           *COEFL(I,J,IP,JP,PP,KK1,KK2,MA)
   98 CONTINUE
   99 CONTINUE
      END
C=================================
      SUBROUTINE SETVV(Z,AMO,NN,NA)
C=================================
C     CALCULATION OF VV
      IMPLICIT REAL*8(A-H,O-Z)
C     COMMON BLOCK FOR VV
      PARAMETER( NV=12 )
      COMMON /COMPOT/VV(0:NV,0:NV)
C
      VV(0,0)=        0
C
      VV(1,0)=        1
C
      VV(2,0)=        1
      VV(2,1)=        0
C
      VV(3,0)=        1
      VV(3,1)=       -3
C
      VV(4,0)=        1
      VV(4,1)=       -8
      VV(4,2)=        0
C
      VV(5,0)=        1
      VV(5,1)=      -15
      VV(5,2)=       20
C
      VV(6,0)=        1
      VV(6,1)=      -24
      VV(6,2)=       75
      VV(6,3)=        0
C
      VV(7,0)=        1
      VV(7,1)=      -35
      VV(7,2)=     -189
      VV(7,3)=     -175
```

```
C
      VV(8,0)=        1
      VV(8,1)=      -48
      VV(8,2)=      392
      VV(8,3)=     -784
      VV(8,4)=        0
C
      VV(9,0)=        1
      VV(9,1)=      -63
      VV(9,2)=      720
      VV(9,3)=    -2352
      VV(9,4)=     1764
C
      VV(10,0)=       1
      VV(10,1)=     -80
      VV(10,2)=    1215
      VV(10,3)=   -5760
      VV(10,4)=    8820
      VV(10,5)=       0
C
      VV(11,0)=       1
      VV(11,1)=     -99
      VV(11,2)=    1925
      VV(11,3)=  -12375
      VV(11,4)=   29700
      VV(11,5)=  -19404
C
      VV(12,0)=       1
      VV(12,1)=    -120
      VV(12,2)=    2904
      VV(12,3)=  -24200
      VV(12,4)=   81675
      VV(12,5)= -104544
      VV(12,6)=       0
C
      DO 10 L=1,NV
      DO 10 K=0,L/2
         VV(L,K)= (0.5D0*NN/AM0)**(L-1)*VV(L,K)
         VV(L,L-K)=  (-1)**(L-1)*VV(L,K)
   10 CONTINUE
C
      VV(1,0)=VV(1,0)-Z
      VV(1,1)=VV(1,1)-Z
      IF(NA.EQ.1) RETURN
```

```
C
      DO 12 L=1,NV
      DO 12 K=0,L
         VV(L,K)=(-1)**(L-1)*VV(L,K)
   12 CONTINUE
      END
C=======================================================
      REAL FUNCTION COEFL*8(IR,IS,IRP,ISP,PP,KK1,KK2,MM)
C=======================================================
C     COEFFICIENT L, OR OVERLAP, < (R,S) (R',S') >
C
      IMPLICIT REAL*8(A-H,O-Z)
C     COMMON BLOCK FOR FACTORIAL
      PARAMETER (NFAC=55)
      COMMON /FACTRL/FAC(0:NFAC)
C
      COEFL=0
C
      IF(IR .LT.-KK1) RETURN
      IF(IRP.LT.-KK1) RETURN
      IF(IS .LT.-KK2) RETURN
      IF(ISP.LT.-KK2) RETURN
C
      SUM=0
      DO 22 I=-1,1
         IF(IR.EQ.IRP+I .AND. IS.EQ.ISP) THEN
            SUM=SUM+COEFA(1,IRP,I,KK1,MM)
         END IF
         IF(IR.EQ.IRP .AND. IS.EQ.ISP+I) THEN
            SUM=SUM+COEFA(1,ISP,I,KK2,MM)
         END IF
   22 CONTINUE
      COEFL=SUM/4/PP**3
     &    *FAC(KK1+IR+MM)/FAC(KK1+IR)
     &    *FAC(KK2+IS+MM)/FAC(KK2+IS)
      END
C=================================================
      REAL FUNCTION COEFT*8(L,I,J,IR,IS,KK1,KK2,MM)
C=================================================
C     COEFFICIENT T
C
      IMPLICIT REAL*8(A-H,O-Z)
C     COMMON BLOCK FOR VV
      PARAMETER( NV=12 )
```

```
      COMMON /COMPOT/VV(0:NV,0:NV)
C
      COEFT=0
C
      IF(L.LT.0) RETURN
      IF(ABS(I).GT.L) RETURN
      IF(ABS(J).GT.L) RETURN
C     IF(IR.LT.-L .OR. IR.GT.L) RETURN
C     IF(IS.LT.-L .OR. IS.GT.L) RETURN
C
      SUM=0
      DO 22 K=0,L
         SUM=SUM+VV(L,K)*COEFA(L-K,IR-I,I,KK1,MM)
     &                  *COEFA(  K,IS-J,J,KK2,MM)
   22 CONTINUE
      COEFT=SUM
      END
C=========================================
      REAL FUNCTION COEFA*8(IP,IR,IS,KK,MM)
C=========================================
C     COEFFICIENTS A
C
      IMPLICIT REAL*8(A-H,O-Z)
C     COMMON BLOCK FOR FACTORIAL
      PARAMETER( NFAC=55 )
      COMMON /FACTRL/FAC(0:NFAC)
C
      COEFA=0
      IF(IS.GT.IP .OR. IR.LT.-KK .OR. IS.LT.MAX(-IP,-KK-IR)) RETURN
C
      IF(IS.GE.0) THEN
         I1=0
         I2=MIN(IP-IS,KK+IR)
      ELSE
         I1=-IS
         I2=MIN(IP,KK+IR)
      END IF
C
      SUM=0
      DO 10 I=I1,I2
         SUM=SUM
     &       +FAC(IP+MM+KK+IR-I)/FAC(I+IS)/FAC(I)
     &       /FAC(IP-IS-I)/FAC(IP-I)/FAC(KK+IR-I)
   10 CONTINUE
```

```
      COEFA=SUM
     &   *(-1)**IS*FAC(IP)**2*FAC(KK+IR+IS)/FAC(KK+IR+IS+MM)
      END
C================================
      SUBROUTINE COEFH(IA,IB,F,I)
C================================
C     COEFFICIENT OF
C     Z**I TERM OF HYPER GEOMETRIC FUNCTION F(IA,IB,Z).
C
      IMPLICIT REAL*8(A-H,O-Z)
      F=1
      A=IA
      B=IB
      DO 10 J=1,I
         F=F*(A+J-1)/(B+J-1)/J
   10 CONTINUE
      END
C======================
      SUBROUTINE SETFAC
C======================
C     FACTORIAL
C
      IMPLICIT REAL*8(A-H,O-Z)
C     COMMON BLOCK FOR FACTORIAL
      PARAMETER( NFAC=55 )
      COMMON /FACTRL/FAC(0:NFAC)
      FAC(0)=1
      DO 10 I=1,NFAC
         FAC(I)=FAC(I-1)*I
   10 CONTINUE
      END
```